
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 12.4.2.3c which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_12_4_2_3c
Test Group:
/12/UTRAN34_GMM_SIMTC.ttcn

ATS Version:
iwd-TTCN3-B2013-03_D16wk37 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7360
Verification Status:
PASS
4 Corrections required for test case 12.4.2.3c
4.1 Introduction

This section describes the changes required to make test case 12.4.2.3c run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D16wk37 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 f_TC_12_4_2_3c_UTRAN()
	Record Name
	f_TC_12_4_2_3c_UTRAN()

	Reason for change
	In the current TTCN implementation of this function, the UE is simply switched ON as per Step 3 of expected sequence in 34.123-1 v 12.4.0 Sec 12.4.2.3c.4. This could potentially mean that the UE may have a P-TMSI, or a TMSI because of last operation. The initial conditions for the testcase (34.123-1 v 12.4.0 Sec 12.4.2.3c.4) clearly state that UE should have a valid IMSI. The current TTCN does not ensure this.
Also, in the current TTCN implementation, the function f_UTRAN34_Postamble(utran34_CellB, U2_CONNECTED); is called for postamble, which assumes that the UE is in connected mode. This is incorrect. After the end of main test body the SS releases the rrc connection. So the UE is in idle mode. The TTCN needs to be corrected accordingly.

	Summary of change
	The functions f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell),f_UTRAN34_AttachReject(utran34_CellA, IMSI),f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE, IMSI_DetachOnly) are called in this specific order to ensure that the UE performs an ATTACH REJ and only has an IMSI as UE identity. 

Also for the postamble called function f_UTRAN34_Postamble(utran34_CellB, U1_IDLE); instead.

Please see screenshot for details.

	Source of change
	UTRAN34_GMM_SIMTC.ttcn

	
	


Before:
	<<SKIPPED CODE>>

// CellA started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    // CellB started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellB);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    f_UTRAN_TestBody_Set(true);
    fl_TC_12_4_2_3c_Body();
    f_UTRAN_TestBody_Set(false);
    //Release UTRAN Cells
    f_UTRAN34_Postamble(utran34_CellB, U2_CONNECTED);
    f_UTRAN_ReleaseCell(utran34_CellA, f_UTRAN_CellInfo_GetConfigType(utran34_CellA));
<<SKIPPED CODE>>


After:

	<<SKIPPED CODE>>

// CellA started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    //WA#12_4_2_3c : Change start
    //Preamble condition to ensure that UE has a IMSI    
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN34_AttachReject(utran34_CellA, IMSI);
    f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE, IMSI_DetachOnly);      
    //WA#12_4_2_3c : Change stop
    // CellB started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellB);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    f_UTRAN_TestBody_Set(true);
    fl_TC_12_4_2_3c_Body();
    f_UTRAN_TestBody_Set(false);
    //Release UTRAN Cells
    f_UTRAN34_Postamble(utran34_CellB, U1_IDLE /*WA#12_4_2_3c U2_CONNECTED*/);
    f_UTRAN_ReleaseCell(utran34_CellA, f_UTRAN_CellInfo_GetConfigType(utran34_CellA));
<<SKIPPED CODE>>


4.3 fl_TC_12_4_2_3b_Body()
	Record Name
	fl_TC_12_4_2_3b_Body()

	Reason for change
	1) In the current TTCN implementation, the ATTACH REQ expected as per Step 5 of the test sequence in 34.123-1 v12.4.0 Sec 12.4.2.3c.4, expects RAI-1. Since the UE has performed ATTACH REJ in preamble, the UEwont be sending the RAI-1 information. This needs to be corrected in current TTCN.
2) The current TTCN implementation of ATTACH ACCEPT message (Step 9 of expected sequence as per 34.123-1 v12.4.0 Sec 12.4.2.3c.4) calls the function f_UTRAN34_AttachAccept. However, no parameter for IE MobileIdentity is passed, which results in the default assignment of TMSI to this IE. This needs to be changed to P-TMSI, since the UE is being assigned a P-TMSI in the attach accept, and has no TMSI as a result of preamble. This needds to be corrected.

3) The current TTCN implementation does not recdonfigure Cells A and B after the rrc conenction is released at Step 11. This results in invalid configuration for Cell B when the Cell is changed to serving cell power and the UE tries to camp on it. This needs to be addressed.
4) The current TTCN expects the RAU REQ at Step 14 as an UplinkDirectTransfer message. This is incorrect since this message is expected on a new rrc connection, and is the first NAS message. This needs to be changed to InitialDirectTransfer message.
5) The current TTCN does not implement the RAU ACCEPT message content as per the Prose at Step 16. The Prose clearly indicates to include P-TMSI which is not being included in current TTCN. Also, the IE GMM Cause is sent with IEI set as omit. This is wrong since it’s a TLV type IE. This also needs to be corrected.
6) The current TTCN expects the RAU REQ at Step 20 as a UplinkDirectTransfer message. This is incorrect and should be expected as a InitialDirectTransfer message. Also, the Update type is expected as Combined RA/LA Updating. This is incorrect. As per 24.008 Sec 4.2.5.1.7, upon expiry of T3302 the UE will initiate RAU with Update Type as “Combined RA/LA updating with IMSI attach”. This needs to be corrected in TTCN. A Prose CR will be raised for this.
7) The current TTCN implementation of RAU ACCEPT message at Step 23 incorrectly sets the IE MSIdentity. The IEI is not being set. This needs to be corrected.

	Summary of change
	1) Instead of checking for RAI in Step 5 against, v_RAI_1, checked for ?.
2) Passed cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def) as a parameter in the ATTACH ACCEPT at Step 9.
3) Called functions f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA) after Step 11. Please see screenshot. 
4) Updated the TTCN to expect the RAU REQ as a InitialDirectTransfer message. Please see screenshot.
5) Passed cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def) as a parameter in the RAU accept message. Also assigned  v_RAUAccept.routingAreaUpdateAccept.gmmCause := cs_GMM_Cause(tsc_GMM_Cause_Congestion, '25'O); instead of v_RAUAccept.routingAreaUpdateAccept.gmmCause := cs_GMM_Cause(tsc_GMM_Cause_Congestion);. Please see screenshot.

6) Updated the TTCN to expect the RAU REQ as a InitialDirectTransfer message. Also changed the expected update type to Combined RA/LA updating with IMSI attach.
7) Assigned the IEI for the Ie MSIdentity to ‘23’O in RAU ACCEPT message sent at Step 23. Please see screenshot.

	Source of change
	UTRAN34_GMM_SIMTC.ttcn

	
	


Before:
	function fl_TC_12_4_2_3c_Body() runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd;
    var template (value) RRCDataReqType v_RAUAccept;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity_A;
    var PLMN_Identity v_PLMN_Identity_B;
    var octetstring v_PLMN_A;
    var octetstring v_PLMN_B;
    var octetstring v_LAC_A;
    var octetstring v_LAC_B;
    var octetstring v_RAC_A;
    var octetstring v_RAC_B;
    var template (present) RoutingAreaId v_RAI_1;
    var template (present) RoutingAreaId v_RAI_4_RX;
    var template (value) RoutingAreaId v_RAI_4_TX;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var InitialUE_Identity v_InitialUE_Id;
    timer t_T3302Ext_UpperBound := f_UG_SetTimerToleranceMax(60.0 * 3.0); // 3 minutes + 10 % tolerance
    timer t_T3302Ext_LowerBound := f_UG_SetTimerToleranceMin(60.0 * 3.0); // 3 minutes - 10 % tolerance
    v_PLMN_Identity_A := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_A);
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_PLMN_Identity_B := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB);
    v_PLMN_B := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_B);
    v_LAC_B := f_UTRAN_CellInfo_GetLAC(utran34_CellB);
    v_RAC_B := f_UTRAN_CellInfo_GetRAC(utran34_CellB);
    v_RAI_1 := cr_RAI(omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    v_RAI_4_RX := cr_RAI(omit, cr_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_RAI_4_TX := cs_RAI(omit, cs_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def);
    // Set the cell type of cell A to the "Serving cell"
    // Set the cell type of cell B to the "Non-Suitable cell"
    // The UE is set in UE operation mode A (see ICS)
    // The UE is powered up or switched on and initiates an attach (see ICS). Cell A is preferred by the UE
    //@siclog "Step 1-3" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    f_UTRAN34_SetUE_OperationMode(opModeA);
    //The UE is powered up or switched on and initiates an attach
    f_UT_SwitchOnUE(UT, false);
    // For Non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    //@siclog "Step 4" siclog@
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 5" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach'
    //Mobile identity = IMSI;
    //TMSI status = no valid TMSI available
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          f_Imsi2MobileIdentity(px_IMSI_Def),
                                                                                          v_RAI_1))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 6-7" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 8" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 9-10 siclog@
    // ATTACH ACCEPT:
    // Attach result = 'Combined GPRS/IMSI attached'
    // Allocated P-TMSI = P-TMSI-2
    // P-TMSI Signature = P-TMSI-2 signature
    // Routing area identity = RAI-1
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeCombined,
                           tsc_PTMSI_2,
                           tsc_PTMSISig_2);
    //@siclog "Step 11" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Steps 12" siclog@
    //Set the cell type of cell A to the "Suitable neighbour cell"
    //Set the cell type of cell B to the "Serving cell"
    //Cell B is preferred by the UE
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);
    //@siclog "Steps 13" siclog@
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 14" siclog@
    // ROUTING AREA UPDATE REQUEST
    // Update type = 'Combined RA/LA updating'
    // Old P-TMSI signature=P-TMSI-2 signature
    // Old Routing area identity = RAI-1
    // TMSI status = no valid TMSI available
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRACombined), //@sic R5s150682 sic@
                                                                 v_RAI_1,
                                                                 cr_PTMSI_Signature(tsc_PTMSISig_2)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    //@siclog "Step 15" siclog@
    // SS starts integrity protection
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 16" siclog@
    // ROUTING AREA UPDATE ACCEPT
    // Update result = 'RA updated'
    // Mobile identity = P-TMSI-1
    // P-TMSI-1 signature
    // Routing area identity = RAI-4
    // GMM cause= 'congestion'
    // T3302= 3min
    v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResultRA_Updated,
                                   v_RAI_4_TX,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   omit, //No new mobile identity
                                   omit);
    v_RAUAccept.routingAreaUpdateAccept.gmmCause := cs_GMM_Cause(tsc_GMM_Cause_Congestion);
    v_RAUAccept.routingAreaUpdateAccept.t3302Value := cs_GPRS_Timer2_3_IEI('2A'O, tsc_GprsTimerUnit_1min, '00011'B); //T3302= 3min // @sic R5s150329 Baseline Moving sic@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));
    //@siclog "Step 17" siclog@
    // ROUTING AREA UPDATE COMPLETE
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
    //Start timers to verify if the UE waits for 3 min before initiating a new routing area update procedure.
    t_T3302Ext_LowerBound.start;
    t_T3302Ext_UpperBound.start;
    //@siclog "Step 18" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    //@siclog "Step 19" siclog@
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellB,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 19");
        }
      [] t_T3302Ext_LowerBound.timeout
        {
          alt {
            [] U_TM.receive(car_RRC_ConnReq(utran34_CellB,
                                            tsc_RB0,
                                            cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
                {
                  t_T3302Ext_UpperBound.stop;
                  v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
                  f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellB, v_InitialUE_Id);
                }
            [] t_T3302Ext_UpperBound.timeout
              {
                //upper bound timer time-out
                f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 19");
              }
          }
        }
    }
    //@siclog "Step 20" siclog@
    // ROUTING AREA UPDATE REQUEST
    // Update type = 'Combined RA/LA updating'
    // Old P-TMSI signature=P-TMSI-1 signature
    // Old Routing area identity = RAI-4
    // TMSI status = no valid TMSI available
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRACombined), //@sic R5s150682 sic@
                                                                 v_RAI_4_RX,
                                                                 cr_PTMSI_Signature(px_PTMSI_SigDef)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.4.2.3c, Step 20");
    //@siclog "Step 22" siclog@
    // SS starts integrity protection
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 23" siclog@
    // ROUTING AREA UPDATE ACCEPT
    // Allocated P-TMSI = P-TMSI-1
    // P-TMSI-1 signature
    // MS identity = IMSI
    // Routing area identity = RAI-4
    v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResultCombRA_LA, //Combined RA/LA updated
                                   v_RAI_4_TX,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def), //Allocated P-TMSI = P-TMSI-1
                                   omit);
    v_RAUAccept.routingAreaUpdateAccept.msIdentity := f_Imsi2MobileIdentity(px_IMSI_Def);
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));
    //@siclog "Step 24" siclog@
    // ROUTING AREA UPDATE COMPLETE
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
    //@siclog "Step 25" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
  }


After:

	function fl_TC_12_4_2_3c_Body() runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd;
    var template (value) RRCDataReqType v_RAUAccept;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity_A;
    var PLMN_Identity v_PLMN_Identity_B;
    var octetstring v_PLMN_A;
    var octetstring v_PLMN_B;
    var octetstring v_LAC_A;
    var octetstring v_LAC_B;
    var octetstring v_RAC_A;
    var octetstring v_RAC_B;
    var template (present) RoutingAreaId v_RAI_1;
    var template (present) RoutingAreaId v_RAI_4_RX;
    var template (value) RoutingAreaId v_RAI_4_TX;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var InitialUE_Identity v_InitialUE_Id;
    timer t_T3302Ext_UpperBound := f_UG_SetTimerToleranceMax(60.0 * 3.0); // 3 minutes + 10 % tolerance
    timer t_T3302Ext_LowerBound := f_UG_SetTimerToleranceMin(60.0 * 3.0); // 3 minutes - 10 % tolerance
    v_PLMN_Identity_A := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_A);
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_PLMN_Identity_B := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB);
    v_PLMN_B := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_B);
    v_LAC_B := f_UTRAN_CellInfo_GetLAC(utran34_CellB);
    v_RAC_B := f_UTRAN_CellInfo_GetRAC(utran34_CellB);
    v_RAI_1 := cr_RAI(omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    v_RAI_4_RX := cr_RAI(omit, cr_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_RAI_4_TX := cs_RAI(omit, cs_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def);
    // Set the cell type of cell A to the "Serving cell"
    // Set the cell type of cell B to the "Non-Suitable cell"
    // The UE is set in UE operation mode A (see ICS)
    // The UE is powered up or switched on and initiates an attach (see ICS). Cell A is preferred by the UE
    //@siclog "Step 1-3" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    f_UTRAN34_SetUE_OperationMode(opModeA);
    //The UE is powered up or switched on and initiates an attach
    f_UT_SwitchOnUE(UT, false);
    // For Non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    //@siclog "Step 4" siclog@
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 5" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach'
    //Mobile identity = IMSI;
    //TMSI status = no valid TMSI available
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          f_Imsi2MobileIdentity(px_IMSI_Def),
                                                                            /*WA#12_4_2_3c v_RAI_1*/ ? ))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }    
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 6-7" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 8" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 9-10 siclog@
    // ATTACH ACCEPT:
    // Attach result = 'Combined GPRS/IMSI attached'
    // Allocated P-TMSI = P-TMSI-2
    // P-TMSI Signature = P-TMSI-2 signature
    // Routing area identity = RAI-1
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeCombined,
                           tsc_PTMSI_2,
                           tsc_PTMSISig_2,
                           cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def));                         
    //@siclog "Step 11" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //@siclog "Steps 12" siclog@
    //Set the cell type of cell A to the "Suitable neighbour cell"
    //Set the cell type of cell B to the "Serving cell"
    //Cell B is preferred by the UE
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);//WA#12_4_2_3c
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);    
    //@siclog "Steps 13" siclog@
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellB, registration);
    //@siclog "Step 14" siclog@
    // ROUTING AREA UPDATE REQUEST
    // Update type = 'Combined RA/LA updating'
    // Old P-TMSI signature=P-TMSI-2 signature
    // Old Routing area identity = RAI-1
    // TMSI status = no valid TMSI available
    U_Dc.receive(/*WA#12_4_2_3c car_PS_UplinkDirectTransfer*/
    


   car_PS_InitDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRACombined), //@sic R5s150682 sic@
                                                                 v_RAI_1,
                                                                 cr_PTMSI_Signature(tsc_PTMSISig_2)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    //@siclog "Step 15" siclog@
    // SS starts integrity protection
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 16" siclog@
    // ROUTING AREA UPDATE ACCEPT
    // Update result = 'RA updated'
    // Mobile identity = P-TMSI-1
    // P-TMSI-1 signature
    // Routing area identity = RAI-4
    // GMM cause= 'congestion'
    // T3302= 3min
    v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResultRA_Updated,
                                   v_RAI_4_TX,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def) /*WA#12_4_2_3c omit*/, //No new mobile identity
                                   omit);
    v_RAUAccept.routingAreaUpdateAccept.gmmCause := cs_GMM_Cause(tsc_GMM_Cause_Congestion, '25'O /*WA#12_4_2_3c*/);
    v_RAUAccept.routingAreaUpdateAccept.t3302Value := cs_GPRS_Timer2_3_IEI('2A'O, tsc_GprsTimerUnit_1min, '00011'B); //T3302= 3min // @sic R5s150329 Baseline Moving sic@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));    
    //@siclog "Step 17" siclog@
    // ROUTING AREA UPDATE COMPLETE
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
    //Start timers to verify if the UE waits for 3 min before initiating a new routing area update procedure.
    t_T3302Ext_LowerBound.start;
    t_T3302Ext_UpperBound.start;
    //@siclog "Step 18" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    //@siclog "Step 19" siclog@
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellB,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 19");
        }
      [] t_T3302Ext_LowerBound.timeout
        {
          alt {
            [] U_TM.receive(car_RRC_ConnReq(utran34_CellB,
                                            tsc_RB0,
                                            cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
                {
                  t_T3302Ext_UpperBound.stop;
                  v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
                  f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellB, v_InitialUE_Id);
                }
            [] t_T3302Ext_UpperBound.timeout
              {
                //upper bound timer time-out
                f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 19");
              }
          }
        }
    }    
    //@siclog "Step 20" siclog@
    // ROUTING AREA UPDATE REQUEST
    // Update type = 'Combined RA/LA updating'
    // Old P-TMSI signature=P-TMSI-1 signature
    // Old Routing area identity = RAI-4
    // TMSI status = no valid TMSI available
    U_Dc.receive(/*WA#12_4_2_3c car_PS_UplinkDirectTransfer*/
    


   car_PS_InitDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, 
                                     /*WA#12_4_2_3c tsc_I_UpdateTypeRACombined*/ tsc_I_UpdateTypeCombinedWithIMSI), //@sic R5s150682 sic@
                                                                 v_RAI_4_RX,
                                                                 cr_PTMSI_Signature(px_PTMSI_SigDef)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.routingAreaUpdateRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.4.2.3c, Step 20");
    //@siclog "Step 22" siclog@
    // SS starts integrity protection
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    f_UTRAN_GMM_Authentication();
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 23" siclog@
    // ROUTING AREA UPDATE ACCEPT
    // Allocated P-TMSI = P-TMSI-1
    // P-TMSI-1 signature
    // MS identity = IMSI
    // Routing area identity = RAI-4
    v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResultCombRA_LA, //Combined RA/LA updated
                                   v_RAI_4_TX,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def), //Allocated P-TMSI = P-TMSI-1
                                   omit);
    v_RAUAccept.routingAreaUpdateAccept.msIdentity := f_Imsi2MobileIdentity(px_IMSI_Def);
    v_RAUAccept.routingAreaUpdateAccept.msIdentity.iei := '23'O; //WA#12_4_2_3c
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));
    //@siclog "Step 24" siclog@
    // ROUTING AREA UPDATE COMPLETE
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
    //@siclog "Step 25" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
  }


4.4 TC_12_4_2_3c
	Record Name
	TC_12_4_2_3c

	Reason for change
	In the current TTCN implementation, the guard timer is set as 300s. This is insufficient given the fact that the testcase contains a timer of 3 minutes, and also involves Switch ON/OFF. The guard timer needs to be extended.

	Summary of change
	Extended the guard timer to 420s.

	Source of change
	UTRAN_Testsuite.ttcn

	
	


Before:
	testcase TC_12_4_2_3c() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Combined routing area updating / congestion / GPRS services only
    var UTRAN_IRAT_PTC   v_UTRAN      := null;
    var GERAN_PTC        v_GERAN      := null;
    var EUTRA_PTC        v_EUTRA      := null;
    timer t_GuardTimer := int2float(300); 
    v_UTRAN := UTRAN_IRAT_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(system, v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_4_2_3c_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop_UTRAN(t_GuardTimer);
  }


After:

	testcase TC_12_4_2_3c() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Combined routing area updating / congestion / GPRS services only
    var UTRAN_IRAT_PTC   v_UTRAN      := null;
    var GERAN_PTC        v_GERAN      := null;
    var EUTRA_PTC        v_EUTRA      := null;
    timer t_GuardTimer := int2float(420); //WA#12_4_2_3c 300->420
    v_UTRAN := UTRAN_IRAT_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(system, v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_4_2_3c_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop_UTRAN(t_GuardTimer);
  }


5 Branches executed in test case 12.4.2.3c
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Intel 7360 device
The Intel 7360 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_12_4_2_3c_Intel_7360.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References
	[1]
	R5s160977
This archive comprises Execution log files, PICS/PIXIT files


